A. Veneziani – Programma per la misura dei riflessi in Visual Basic

Premessa

In questa breve dispensa mi occuperò di descrivere il sistema per la misura dei riflessi, già presentato nelle sue linee generali qualche lezione fa.   Il sistema si pone, come obbiettivo,  quello di misurare quanto sia rapido il riflesso di una persona dato un certo stimolo noto (nel nostro caso passaggio di un’area della form dal colore verde al colore rosso).  Per la misura dei tempi necessari si è fatto ricorso al componente Timer appositamente proposto in Visual Basic.

Alcuni dettagli del progetto 

La misura abbisogna, almeno nel nostro caso, di alcuni timer, uno dei quali direttamente preposto a misurare quanto tempo sia passato quando l’utente reagisce, da quando il colore della apposita zona è cambiato (da verde a rosso). 

La reazione dell’utente viene rilevata tramite la pressione di un pulsante sulla form.  In una apposita label appare il tempo trascorso da quando c’è stato il cambiamento di colore, a quando l’utente reagisce al cambiamento premendo il pulsante.  Il tempo, nel listato in esame, è stato approssimato in slot minimi di 10 millisecondi, ossia un centesimo di secondo.

La misura  del tempo di reazione viene visualizzata in una apposita label che indica il tempo di reazione risultante per quel tentativo.  Un’altra label in altro momento riporta quanti tentativi siano già stati fatti.      Successivamente al cambiamento di colore dal verde al rosso, parte un timer che conta qualche secondo (nel nostro caso 2), e che successivamente comanda di riportare al verde il segnalatore per una nuova misura di riflesso.  In pratica questo timer definisce un tempo ben definito tra una prova e l’altra di riflessi.  Terminato il suo periodo questo timer riporta la zona colorata a verde.  Viene in questo momento attivato un timer con intervallo di ritardo legato ad un numero casuale.

Il passaggio dal colore verde al colore rosso viene infatti regolato dal timer di cui sopra, in modo casuale, per evitare che l’utente possa aspettarsi il momento in cui la zona colorata cambi colore (essendo il tempo variabile, ciò non viceversa è possibile, ossia il cambiamento è sempre non prevedibile, in quanto casuale).

La sequenza del funzionamento dei 3 timer impiegati si può quindi schematizzare come:

Timer (2)  casuale ritardo del rosso                 Timer (3)  rilevamento tempo di reazione


Timer (1)  ritardo ritorno al verde 

L’interfaccia del programma è qui mostrata durante un momento del suo funzionamento.

Si notano facilmente gli elementi visibili sull’ interfaccia stessa, ossia la label colorata, il pulsante che permette il rilevamento del riflesso, l’indicatore del tempo trascorso ossia di quanto il riflesso avuto è stato pronto, e l’etichetta che indica a che tentativo siamo nel rilevare riflessi.

In realtà la form ospita anche tre controlli timer,a run time nascosti, come tutti i controlli timer, che permettono le temporizzazioni di cui si è parlato sopra.  E’ da notare come fatto in altre dispense che il controllo timer è un controllo che permette, seppur sotto certe condizioni di inserire la concorrenza in Visual Basic, ed infatti come si può notare anche dal diagramma temporale i timer 1 e 3 lavorano in parallelo, ossia contano il tempo in modo indipendente l’uno dall’altro e contemporaneamente.

Variante con registrazione di alcuni dati statistici

Il programmino che abbiamo commentato si presta bene a essere variato in vari modi;  uno dei possibili è proporre che si vada a registrare tutti i risultati in un array dei tempi di ritardo (tempo di ritardo = tempo passato dall’apparizione del rosso al momento in cui l’utente preme il pulsante di reazione), così da poter rilasciare all’occorrenza dati statistici che ne abbisognano.

Per far ciò viene utilizzato nel listato un array di numeri che permette di memorizzare tutti i tempi di ritardo rilevati via via.    In questo modo è poi possibile, ad esempio,  determinare quale sia stato il tempo massimo negli ultimi test o il tempo minimo o calcolare la media e altro.

In questo caso il programma, come mostra la figura qui a fianco possiede un pulsante in più atto a comandare l’apertura di una messagebox che riporti tali dati.

Il componente Timer

Il componente Timer è uno dei componenti “nativi” (ossia quelli standard, disponibili all’avvio del programma) di Visual Basic.  Le proprietà fin qui utilizzate del componente, che sono poi quelle effettivamente rilevanti, sono:

Enabled  ->  indica se il timer stà conteggiando (true) o meno (false) del tempo

Interval ->  passo col quale viene richiamata la routine evento del timer Timer_Timer()

Il funzionamento di un timer, come detto in altre occasioni, si concretizza in una chiamata ripetuta della rouitne evento ad intervalli regolari, la cui distanza l’uno dall’altro è data dal valore della proprietà Interval, che viene espresso in millisecondi.  Ovviamente altra proprietà importante è Name, sempre presente del resto in qualunque controllo.  Per ulteriori informazioni su Timer, si consulti libro di testo e documentazione Visual Basic abbinata al programma.





                    Eventi  Timer()
Commento al listato

Riporto qui un essenziale commento al listato Visual Basic.   Il listato presentato è peraltro già commentato sinteticamente , come d’uso.   Il listato che vado a prendere in considerazione è quello della versione base del programma di rilevamento di riflessi.

Nell’evento Form_Load troviamo varie operazioni che attengono o vengono eseguite all’avvio del programma:

Private Sub Form_Load()

    Tentativi = 0     ' azzera conteggio tentativi

    Randomize           ' inizializza generatore casuale per produrre effettivamente random

    Ritardo = Rnd * 10     ' crea un numero entro 10 (sec.)

    Ritardo = Int(Ritardo * 1000)   ' (converte il tutto in ms.)

    Timer1.Enabled = True       ' attiva il timer che genera i *** ritardi variabili del rosso ***

    Timer1.Interval = Ritardo     ' imposta il ritardo su Timer1

End Sub

Inizialmente troviamo che viene azzerato il contatore dei tentativi,  viene inizializzato il generatore di numeri casuali (istruzione Randomize) , viene generato un primo ritardo, e viene attivato il relativo timer (timer1) per contare il ritardo di cui detto.   Scaduto il suo tempo (variabile a caso) timer1attiva la sua relativa routine evento che è così descritta:

Private Sub Timer1_Timer()

    Timer2.Enabled = True

    Timer3.Enabled = True

    Timer1.Enabled = False

    ' cambia l'etichetta da verde a rossa

    Label1.BackColor = vbRed

    Tentativi = Tentativi + 1        ' registra che un nuovo tentativo è in corso

    Label2.Caption = "Tentativo n°:  " & Tentativi   ' aggiorna la label cheindica a che tentativo siamo

    txt_DeltaTime.Text = ""

End Sub

Trascorso quindi il tempo previsto vengono abilitati i timer2 e 3 rappresentanti la misura del ritardo e viene aggiornata la label che segna il numero di tentativo in corso.   Viene invece ripulita la segnalazione di quanto tempo era stato necessario per il riflesso precedente (misura del riflesso).  Viene anche cambiato colore alla label colorata portandolo a rosso  (Label1.BackColor = vbRed).

Mentre i due timer sono attivi succede che:  timer3 richiama più e è più volte, a brevi intervalli di tempo, la routine evento Timer3_Timer().   Questa routine evento incrementa solo una variabile intera di conteggio.  La routine cessa di attivarsi solo quando viene premuto il pulsante che segnala la registrazione di un riflesso utente. Tale valore moltiplicato per il valore della proprietà Interval del timer indicherà il tempo totale passato prima del riflesso dell’utente (ovviamente dal momento del passaggio al rosso della label colorata).

L’altro timer (timer2) viene attivato solo una volta alla fine del tempo previsto per la permanenza del rosso nella label colorata.  Finito questo intervallo,  viene ricalcolata dalla stessa routine evento di timer2 un valore casuale di secondi e vengono assegnati come nuovo valore della proprietà Interval di timer1.   Timer1 come prima detto è il timer che lavora con valore di tempo casuale.  Viene poi attivato Timer1 (che era “spento”), e vengono invece stoppati i timer 2 e 3 appena utilizzati.   Infine viene rimesso a verde il colore della label colorata.  Infatti bisogna tenere presente che questo timer ha la funzione di realizzare un ritardo fisso nel ritornare del rosso al verde.

Private Sub Timer2_Timer()

    ' quando termina il periodo in cui è visibile il rosso

    ' imposta i parametri di ritardo per un nuovo tentativo

    Ritardo = Rnd * 10

    Ritardo = Int(Ritardo * 1000)

    ' attiva il timer per il ritardo casuale del rosso (Timer1) e spegne Timer2

    ' che ha terminato, per il momento il suo compito

    Timer1.Enabled = True

    Timer1.Interval = Ritardo

    Timer2.Enabled = False

    Timer3.Enabled = False   ' stoppa il timer 3 (conteggio del tempo passato sul rosso)

    NumTick = 0                 ' azzera il conteggio (la label e' attualmente di nuovo verde)

    Label1.BackColor = vbGreen     ' fà riapparire, ovviamente, il verde

End Sub

Viene anche riazzerato il conteggio dei colpi di clock (contati dalla variabile NumTick) effettuati da timer3, in base al quale viene poi effettivamente conteggiato quanto sia stato, o meno, pronto il riflesso dell’utente, ossia il tempo di ritardo.

In definitiva, come si nota dal grafico temporale proposto prima, i timer1 e 2 lavorano in modo alternato l’uno all’altro.  Quando uno conta il tempo l’altro è in riposo e viceversa.   In questo modo viene realizzato un ciclo piuttosto particolare, si potrebbe dire virtuale, in cui le due routine evento di timer1 e 2 si attivano in maniera alternata.  Al momento in cui si attiva timer2 si attiva in modo parallelo anche timer3, per il conteggio del tempo di ritardo.

Il tempo di ritardo viene definito al momento che viene l’apposito pulsante sulla form, la cui routien evento corrispondente è:

Private Sub cmd_Intercetta_Click()

    Timer3.Enabled = False         ' stoppa Timer3 ossia il conteggio del ritardo

    txt_DeltaTime.Text = NumTick * Timer3.Interval & "  ms."   ' dal numero di tick contati deduce il tempo di ritardo

    NumTick = 0  ' azzera conteggio tempo Timer3

End Sub

Come si può osservare questa routine evento ferma il timer3 immediatamente, fa un rapido calcolo del tempo conteggiato da timer3, e lo visualizza in una apposita label e successivamente azzera il numero di colpi di clock contati in questo tentativo di misura del riflesso.

Questo è tutto quello che è necessario osservare e sapere sul funzionamento del programma-base.

L’analogo programma con capacità statistiche ha solo poche variazioni, come detto, tra cui la presenza di un array di numeri (array Memo), per permettere di memorizzare via via i risultati (tempi di ritardo);  questa memorizzazione, come ovvio avviene nel momento in cui il tempo di ritardo viene definito, ossia in Sub cmd_Intercetta_Click()  Inoltre gli appositi calcoli e l’attivazione della MessageBox avvengono nella routine dell’apposito pulsante fatto per attivare la visualizzazione di tali dati.

Il contenuto dell’intera routine evento, per avere un idea di insieme è:

Private Sub cmd_Stat_Click()

Dim Cc As Integer

Dim TmMin As Integer, TmMax As Integer, TmMed As Integer

Dim Somma As Integer, Tentativi_Validi As Integer

    TmMin = 32767              ' siccome deve calare deve partire dal valore piu' alto possibile

    Tentativi_Validi = 0          ' conta il numero di tentativi di prova-riflessi effettivamente effettuati

    For Cc = 1 To Tentativi          ' ciclo di scansione di tutti i dati raccolti

        ' Individua il minimo della serie di dati

        If (TmMin > Memo(Cc)) And (Memo(Cc) <> 0) Then TmMin = Memo(Cc)

        ' Individua il massimo della serie di dati

        If TmMax < Memo(Cc) Then TmMax = Memo(Cc)

        ' Conta i tenativi validi effettuati (per calcolo media)

        If Memo(Cc) <> 0 Then Tentativi_Validi = Tentativi_Validi + 1

        ' Conta il totale dei tempi di ritardo (per calcolo media)

        Somma = Somma + Memo(Cc)

    Next

    ' calcolo del tempo medio di ritardo

    If (Tentativi_Validi <> 0) Then TmMed = Somma / Tentativi_Validi

    ' Pone una indicazione 0 se non esistono ancora tempi minimi rilevati...

    If TmMin = 32767 Then TmMin = 0

    ' esprime i dati calcolati in una message box (se il dato non e' presente  viene data una indicazione particolare)

    MsgBox "Tempo più alto: " & TmMax & "    Tempo minimo:  " & TmMin & "   Tempo medio dei tentativi:  " & TmMed, _

            vbInformation + vbOKOnly, "Statistiche attuali"

End Sub

Il valore minimo e massimo vengono, ad esempio, determinati così:


…     

' Individua il minimo della serie di dati

If (TmMin > Memo(Cc)) And (Memo(Cc) <> 0) Then TmMin = Memo(Cc)

' Individua il massimo della serie di dati

If TmMax < Memo(Cc) Then TmMax = Memo(Cc)


...

Questo frammento di codice è interno ad un ciclo For che scandisce i vari valori rilevati.  Sempre tramite questo ciclo viene calcolata la somma dei tempi di ritardo, necessaria per calcolare la media degli stessi.

Questo invece è il comando che fa apparire la box di messaggio, con i dati (tempo di risposta minimo, massimo e medio):

' esprime i dati calcolati in una message box (se il dato non e' presente  viene data una indicazione particolare)

    MsgBox "Tempo più alto: " & TmMax & "    Tempo minimo:  " & TmMin & "   Tempo medio dei tentativi:  " & TmMed, _

            vbInformation + vbOKOnly, "Statistiche attuali"

Sempre nella stessa routine evento, le indicazioni 32767 stanno ad indicare il massimo numero intero ossia quella che in altri linguaggi viene indicata come la costante MAXINT (intero a 16 bit).

